
American Airlines stock volume – 25 years:  
 

 
Near 9/11/2001 

 



 
Vertical lines at 9/11/2001 and 12/11/2001 (after crash off 
of NY) 
Black “predictions” from exponentially weighted moving 
average with  11 1.ˆ9.ˆ

−− += ttt YYY .  Time series model may help. 
9/11 and next crash intervention model (SAS)  
 
axis1 label = (angle = 90);   
 
************************************* 
*  Input and sort the data          * 
************************************; 
Data AMR; title "Large AMR dataset"; 
infile "F:\SAMSI\Americanbig.csv" dlm="," dsd firstobs=2 missover;  
length datedum $ 9.;  
input datedum $  open high low close volume;  
date = input(datedum, date9.);  drop datedum;  
format date date7.; 
proc sort data=AMR; by date;  
proc gplot; plot volume*date; symbol1 v=none i=join;  
run;  
 
******************************** 
* Add intervention variables   * 
*******************************; 
Data American; set AMR;  
WTC = (date="17sep2001"d); 
CRASH = (date="12nov2001"d);  
label CRASH = "Crash of AA Flt. 587 east of New York"; 
surprise = volume - ima;  
if _n_ = 1 then IMA = volume;  
else IMA = .9*IMA+.1*volume;  
format date date7.; retain IMA; 
 
************************ 
* Graph the data       * 
***********************; 
proc gplot; Title "Simple EWMA";  
plot (ima volume)*date/overlay href="17sep2001"d "12nov2001"d  
chref=cyan vaxis=axis1;  
where '01jan01'd < date < '01jan03'd; 
symbol1 v=none i=join;  
proc print; var date volume; 
where ('01sep01'd < date < '01jan03'd)and(volume>5000000); 
run;  
 
******************************  
*  Fit intervention model    * 
*****************************; 
proc arima data=AMERICAN;  



i var=volume crosscor=(WTC CRASH) noprint;  
e input = ( (1)/(1,2) WTC (1)/(1,2) CRASH) plot p=2 q=1;   
f lead=0 out=out1 id=date;  
where '01jan01'd < date < '01jan03'd; 
proc print data=out1 (obs=10);  
 
***************************** 
*  Plot results             * 
****************************; 
 
proc gplot data=out1; title "ARIMA transfer function"; 
plot (forecast volume)*date/overlay href="17sep2001"d "12nov2001"d  
chref=cyan vaxis=axis1;  
where '01jan01'd < date < '01jan03'd; 
proc gplot data=out1; title "ARIMA transfer function"; 
plot (forecast volume)*date/overlay vaxis=axis1;  
plot residual*date/overlay vref=0 vaxis=axis1;  
where '01jul01'd < date < '11sep01'd; 
proc gplot data=American; title "EWMA 'surprises'"; 
plot surprise*date/overlay vref=0 vaxis=axis1;  
where '01jul01'd < date < '11sep01'd; 
run; 
********************************************************************** 
 
 
 
 
                         The ARIMA Procedure 
 
                  Conditional Least Squares Estimation 
 

                       Standard            Approx 
Parameter  Estimate       Error  t Value   Pr>|t|  Lag   Variable     
 
MU       1895175.1      241956.3   7.83   <.0001    0    volume 
MA1,1      0.80798       0.06487  12.45   <.0001    1    volume 
AR1,1      1.30707       0.08535  15.31   <.0001    1    volume 
AR1,2     -0.33545       0.07374  -4.55   <.0001    2    volume 
NUM1      15624717      831551.1  18.79   <.0001    0    WTC    
NUM1,1  -9564494.0     3086912.7  -3.10   0.0021    1    WTC     
DEN1,1    -0.21308       0.18425  -1.16   0.2481    1    WTC      
DEN1,2     0.38677       0.07282   5.31   <.0001    2    WTC   
NUM2     7534484.3      842654.2   8.94   <.0001    0    CRASH  
NUM1,1   4260018.6     2401054.1   1.77   0.0767    1    CRASH 
DEN1,1     0.79740       0.31164   2.56   0.0108    1    CRASH 
DEN1,2     0.04974       0.17505   0.28   0.7764    2    CRASH 

 



Interpretation: 
 
 
Xt = WTC indicator 0 0 0 1 0 0 0 … 

Similarly for Xt = second crash indicator 
Error term is ARMA(2,1)  
 
Residual Checks  
 
 
              Autocorrelation Check of Residuals 
To    Chi-        Pr > 
Lag  Square DF   ChiSq    ---------Autocorrelations-------------- 
 
 6   11.07   3  0.0114  -0.002 -0.008  0.016  0.060  0.059 -0.121  
12   13.69   9  0.1336   0.032 -0.019 -0.042 -0.030 -0.034 -0.002  
18   20.49  15  0.1541  -0.082 -0.017  0.005  0.013 -0.074  0.023 
24   37.27  21  0.0157   0.116  0.088 -0.087 -0.036  0.045  0.011 
30   43.28  27  0.0245  -0.046 -0.016  0.059 -0.041  0.061  0.008 
36   51.68  33  0.0203   0.069  0.013 -0.042  0.085  0.032  0.029 
42   55.44  39  0.0425  -0.045  0.020  0.066 -0.002 -0.011 -0.002 
48   56.92  45  0.1096   0.020 -0.011 -0.009  0.002 -0.005  0.046 

 
 
 
 
 
 
Forecasts from this model shown below:  
 



 
 
 
Data before 9/11with transfer function forecast: 

 



Residuals before 9/11/01:  
 

 
Example 2:   “Nenana Ice Classic”  
 

 
 



Touted as evidence of global warming 
In 1917, engineers at Tanana river near Nenana Alaska 
erected tripod on ice, connect to clock, bet on time tripod 
would move downstream to stop clock -> beginning of 
spring.  Betting caught on and continues each year – pot is 
now $285,000.  Signal is (possibly) ramp intervention.    
 
options ls=76;   ** Demos  Nenana.sas  **;  
* Ice breakup time in Tanana River near Nenana,  
Alaska - for years residents have bet on the  
breakup time, which is considered the unofficial 
start of spring.  
A wooden tripod is used to measure the breakup.  
More information is available at  
 
http://fairbanks-alaska.com/nenana.htm 
 ;  
Data Ice; input year 1-4 mon $ 8-10 day 12-13;  
month = 5; if mon="Apr" then month=4;  
break = MDY(month,day,1960);  
*23456789 123456789 123456789;  
cards;  
1917   Apr 30  11:30 a.m.  
1918   May 11   9:33 a.m.  
1919   May  3   2:33 p.m.  
1920   May 11  10:45 a.m.  
1921   May 11   6:42 a.m.  
1922   May 12   1:20 p.m.  
1923   May  9   2:00 a.m.  
1924   May 11   3:10 p.m.  
1925   May  7   6:32 p.m.  
1926   Apr 26   4:03 p.m.  
1927   May 12   5:42 a.m.  
1928   May  6   4:25 p.m.  
1929   May  5   3:41 p.m.  
1930   May  8   7:03 p.m.  
1931   May 10   9:23 a.m.  
1932   May  1  10:15 a.m.  
1933   May  8   7:30 p.m.  
1934   Apr 30   2:07 p.m.  
1935   May 15   1:32 p.m.  
1936   Apr 30  12:58 p.m.  
1937   May 12   8:04 p.m.  
1938   May  6   8:14 p.m.  
1939   Apr 29   1:26 p.m.  
1940   Apr 20   3:27 p.m.  
1941   May  3   1:50 a.m.  
1942   Apr 30   1:28 p.m.  
1943   Apr 28   7:22 p.m.  
1944   May  4   2:08 p.m.  
1945   May 16   9:41 a.m.  



1946   May  5   4:40 p.m.  
1947   May  3   5:53 p.m.  
1948   May 13  11:13 a.m.  
1949   May 14  12:39 p.m.  
1950   May  6   4:14 p.m.  
1951   Apr 30   5:54 p.m.  
1952   May 12   5:04 p.m.  
1953   Apr 29   3:54 p.m.  
1954   May  6   6:01 p.m.  
1955   May  9   2:13 p.m.  
1956   May  1  11:24 p.m.  
1957   May  5   9:30 a.m.  
1958   Apr 29   2:56 p.m.  
1959   May  8  11:26 a.m.  
1960   May  2   7:12 p.m.  
1961   May  5  11:31 p.m.  
1962   May 12  11:23 p.m.  
1963   May  5   6:25 p.m.  
1964   May 20  11:41 a.m.  
1965   May  7   7:01 p.m.  
1966   May  8  12:11 p.m.  
1967   May  4  11:55 a.m.  
1968   May  8   9:26 a.m.  
1969   Apr 28  12:28 p.m.  
1970   May  4  10:37 a.m.  
1971   May  8   9:31 p.m.  
1972   May 10  11:56 a.m.  
1973   May  4  11:59 a.m.  
1974   May  6   3:44 p.m.  
1975   May 10   1:49 p.m.  
1976   May  2  10:51 a.m.  
1977   May  6  12:46 p.m.  
1978   Apr 30   3:18 p.m.  
1979   Apr 30   6:16 p.m.  
1980   Apr 29   1:16 p.m.  
1981   Apr 30   6:44 p.m.  
1982   May 10   5:36 p.m.  
1983   Apr 29   6:37 p.m.  
1984   May  9   3:33 p.m.  
1985   May 11   2:36 p.m.  
1986   May  8  10:50 p.m.  
1987   May  5   3:11 p.m.  
1988   Apr 27   9:15 a.m.  
1989   May  1   8:14 p.m.  
1990   Apr 24   5:19 p.m.  
1991   May  1   12:04 a.m.  
1992   May 14   6:28 a.m.  
1993   Apr 23   1:01 p.m.  
1994   Apr 29  11:01 p.m.  
1995   Apr 26   1:22 p.m.  
1996   May  5  12:32 p.m.  
1997   Apr 30  10:28 a.m.  
1998   Apr 20   4:54 p.m.  
1999   Apr 29   9:47 p.m.  
2000   May  1  10:47 a.m.  
2001   May  8   1:00 p.m. 
2002   May  7   9:27 p.m. 



2003   Apr 29   6:22 p.m. 
2004   Apr 24   2:16 p.m. 
   ; 
%let yr0=1967;   
data dates; do year=1917 to 2020;  
ramp=(year-&yr0)*(year>&yr0); output; end;  
data all; merge Ice dates; by year; 
proc arima; i var=break crosscor=(ramp) noprint; 
e input=(ramp) p=(6); 
f lead=5 id=year out=out1;  
proc gplot;  
plot (break u95 l95 forecast)*year/overlay href=&yr0;  
title "Tanana River Ice Breakup"; 
symbol1 v=dot i=join c=black;  
symbol2 v=none i=join l=2 c=gray r=2;  
symbol3 v=none i=join c=gray;  
run;  
 
** Optional: Using NLIN to estimate ramp start point **;  
proc nlin data=all;  
parms C=1960 a=126 b=-.2 ;  
  X = (year-C)*(year>c);  
model break = a + b*X;  
run;  
 

 
 



                         The ARIMA Procedure 
                Conditional Least Squares Estimation 
 
                         Standard             Approx 
Parameter    Estimate       Error  t Value  Pr > |t|   Lag  Variable   
MU          126.01962     0.59155   213.03    <.0001     0  break      
AR1,1        -0.21784     0.10929    -1.99    0.0494     6  break      
NUM1         -0.18686     0.04326    -4.32    <.0001     0  ramp       
 
               Autocorrelation Check of Residuals 
  To    Chi-       Pr > 
 Lag  Square DF   ChiSq  -------------Autocorrelations------------- 
 
   6    4.81  5  0.4399   -0.040  0.043 -0.064  0.054 -0.197 -0.034 
  12   14.37 11  0.2134    0.056 -0.089 -0.090  0.215 -0.023 -0.167 
  18   18.06 17  0.3850   -0.044  0.079  0.033 -0.060  0.129 -0.063 
  24   21.22 23  0.5679   -0.054  0.031  0.016 -0.123 -0.036 -0.075 
 
 
                         Model for variable break 
                      Estimated Intercept    126.0196 
 
                           Autoregressive Factors 
                        Factor 1:  1 + 0.21784 B**(6) 
 
                              Input Number 1 
                   Input Variable                   ramp 
                   Overall Regression Factor    -0.18686 
 
                          Tanana River Ice Breakup                          
 
                             The NLIN Procedure 
                          Dependent Variable break 
                            Method: Gauss-Newton 
 
                              Iterative Phase 
                                                          Sum of 
            Iter           C           a           b     Squares 
               0      1960.0       126.0     -0.2000      2850.8 
               1      1975.1       125.6     -0.2248      2714.6 
               2      1965.7       126.0     -0.1527      2710.9 
               3      1967.8       126.0     -0.1860      2702.2 
               4      1967.6       126.0     -0.1873      2702.1 
               5      1967.6       126.0     -0.1873      2702.1 
      NOTE: Convergence criterion met. 
 



                          Sum of        Mean               Approx 
 Source           DF     Squares      Square    F Value    Pr > F 
 
 Model             2       403.5       201.7       6.35    0.0027 
 Error            85      2702.1     31.7894 
 Corrected Total  87      3105.6 
 
                                 Approx   Approximate 95% Confidence 
   Parameter      Estimate    Std Error            Limits 
   C                1967.6      10.7354      1946.2   1988.9 
   a                 126.0       0.7895       124.4    127.6 
   b               -0.1873       0.0868     -0.3599  -0.0147  
 
                    Approximate Correlation Matrix 
                           C               a               b 
           C       1.0000000      -0.3926600      -0.7957997 
           a      -0.3926600       1.0000000       0.0000000 
           b      -0.7957997       0.0000000       1.0000000 

 


